STAT 5200 Handout #8: Factorial Design (Ch. 8)

Example: We have two kinds of Glass (1, 2) and three kinds of Phosphor (A, B, C) [for a total of six Combinations] to use in making TV tubes. Three replicates of each combination are used. For each experimental unit, the Output of the resulting screen is measured (current needed to produce 30 foot-lamberts); lower Output means more efficient brightness, not lower brightness.

```
/* Enter data */
data tvtube;
  input Combo Glass Phosphor $ Output @@;
  label Output = 'Current Needed to Produce 30 foot-lamberts';
  cards;
   1 1 A 280   1 1 A 290   1 1 A 285
   2 1 B 300   2 1 B 310   2 1 B 295
   3 1 C 270   3 1 C 285   3 1 C 290
   4 2 A 230   4 2 A 235   4 2 A 240
   5 2 B 260   5 2 B 240   5 2 B 235
   6 2 C 220   6 2 C 225   6 2 C 230
   ;
run;

/* Look at single-factor CRD */
proc glm data=tvtube plots=diagnostics;
  class Combo;
  model Output = Combo;
  means Combo / regwq HOVtest=Levene;
  title1 'Single-Factor CRD';
run;
```

Single-Factor CRD

Dependent Variable: Output Current Needed to Produce 30 foot-lamberts

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	5	14627.77778	2925.55556	42.13	<.0001
Error	12	833.33333	69.44444		
Corrected Total	17	15461.11111			

Source	DF	Type III SS	Mean Square	F Value	Pr > F
Combo	5	14627.77778	2925.55556	42.13	<.0001



Levene's Test for Homogeneity of Output Variance ANOVA of Squared Deviations from Group Means					
Source	**DF**	**Sum of Squares**	**Mean Square**	**F Value**	**Pr > F**
Combo	5	24938.3	4987.7	1.88	0.1709
Error	12	31759.3	2646.6		

H_0: error variance is constant



Ryan-Einot-Gabriel-Welsch Multiple Range Test for Output

Note: This test controls the Type I experimentwise error rate.

Means with the same letter are not significantly different.			
REGWQ Grouping	**Mean**	**N**	**Combo**
A	301.667	3	2
A			
A	285.000	3	1
A			
A	281.667	3	3
B	245.000	3	5
B			
B	235.000	3	4
B			
B	225.000	3	6

```
/* Now account for factor structure: factorial design */
proc glm data=tvtube plots=diagnostic;
  class Glass Phosphor;
  model Output = Glass Phosphor Glass*Phosphor;
  title1 'Two-Factor (2*3) Factorial Design';
run;
```

*Two-Factor (2*3) Factorial Design*

Dependent Variable: Output Current Needed to Produce 30 foot-lamberts

ANOVA table

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	5	14627.77778	2925.55556	42.13	<.0001
Error	12	833.33333	69.44444		
Corrected Total	17	15461.11111			

partitioned

Source	DF	Type III SS	Mean Square	F Value	Pr > F
Glass	1	13338.88889	13338.88889	192.08	<.0001
Phosphor	2	1244.44444	622.22222	8.96	0.0042
Glass*Phosphor	2	44.44444	22.22222	0.32	0.7322

Sampling distn is $F_{1,12}$

$F_{2,12}$

ave Y when Glass=1 & Phosphor=B

Y



```
/* Look at post-hoc tests and interaction plot;
   note shortcut in MODEL statement to include
   interaction */
proc glm data=tvtube plots=diagnostic;
  class Glass Phosphor;
  model Output = Glass | Phosphor;
  means Glass Phosphor Glass*Phosphor / REGWQ;
  lsmeans Glass*Phosphor / pdiff adjust=tukey lines;
  title1 'Post-hoc tests and interaction plot';
  title2 '(p. 2 of Handout #8b)';
run;
```

```
NOTE: Means from the MEANS statement are not adjusted for other
      terms in the model.  For adjusted means, use the LSMEANS
      statement.
```

(REGWQ only works for single-factor models)

Post-hoc tests and interaction plot
(p. 2 of Handout #8b)

Source	DF	Type III SS	Mean Square	F Value	Pr > F
Glass	1	13338.88889	13338.88889	192.08	<.0001
Phosphor	2	1244.44444	622.22222	8.96	0.0042
Glass*Phosphor	2	44.44444	22.22222	0.32	0.7322





Ryan-Einot-Gabriel-Welsch Multiple Range Test for Output

Note: This test controls the Type I experimentwise error rate.

Means with the same letter are not significantly different.			
REGWQ Grouping	**Mean**	**N**	**Glass**
A	289.444	9	1
B	235.000	9	2



Means with the same letter are not significantly different.			
REGWQ Grouping	**Mean**	**N**	**Phosphor**
A	273.333	6	B
B	260.000	6	A
B			
B	253.333	6	C





Level of Glass	Level of Phosphor	N	Output	
			Mean	Std Dev
1	A	3	285.000000	5.0000000
1	B	3	301.666667	7.6376262
1	C	3	281.666667	10.4083300
2	A	3	235.000000	5.0000000
2	B	3	245.000000	13.2287566
2	C	3	225.000000	5.0000000

Least Squares Means
Adjustment for Multiple Comparisons: Tukey

Glass	Phosphor	Output LSMEAN	LSMEAN Number
1	A	285.000000	1
1	B	301.666667	2
1	C	281.666667	3
2	A	235.000000	4
2	B	245.000000	5
2	C	225.000000	6

Least Squares Means for effect Glass*Phosphor Pr > |t| for H0: LSMean(i)=LSMean(j) Dependent Variable: Output						
i/j	1	2	3	4	5	6
1		0.2139	0.9956	0.0001	0.0008	<.0001
2	0.2139		0.0999	<.0001	<.0001	<.0001
3	0.9956	0.0999		0.0002	0.0017	<.0001
4	0.0001	<.0001	0.0002		0.6879	0.6879
5	0.0008	<.0001	0.0017	0.6879		0.0999
6	<.0001	<.0001	<.0001	0.6879	0.0999	

only signif. differences are Glass 1 vs. Glass 2



Tukey Comparison Lines for Least Squares Means of Glass*Phosphor

LS-means with the same letter are not significantly different.

	Output LSMEAN	Glass	Phosphor	LSMEAN Number
A	301.667	**1**	**B**	2
A	285.000	**1**	**A**	1
A	281.667	**1**	**C**	3
B	245.000	**2**	**B**	5
B	235.000	**2**	**A**	4
B	225.000	**2**	**C**	6

```
/* See other interaction plot */
proc glm data=tvtube;
  class Phosphor Glass;  /* only change is order here */
  model Output = Glass Phosphor Glass*Phosphor;
  title2 'Different order -- same model';
run;
```




```
/* Try a contrast based on 'combination' model */
proc glm data=tvtube;
 class Combo;
 model Output = Combo;
 contrast 'Phosphor A vs. Phosphor C'
    Combo 1 0 -1 1 0 -1;
 title1 '(p. 3 of Handout #8b)';
run;
```

(p. 3 of Handout #8b)

Contrast	DF	Contrast SS	Mean Square	F Value	Pr > F
Phosphor A vs. Phosphor C	1	133.3333333	133.3333333	1.92	0.1911